

11 Cool Syntax

Figure 1 provides a specification of Cool syntax. The specification is not in pure Backus-Naur Form (BNF); for convenience, we also use some regular expression notation. Specifically, A^* means zero or more A 's in succession; A^+ means one or more A 's. Items in square brackets $[..]$ are optional. Double brackets $[[..]]$ are not part of Cool; they are used in the grammar as a meta-symbol to show association of grammar symbols (e.g. $a[[bc]]^+$ means a followed by one or more bc pairs).

11.1 Precedence

The precedence of infix binary and prefix unary operations, from highest to lowest, is given by the following table:

```
.
@
~
isvoid
* /
+ -
<= < =
not
<-
```

All binary operations are left-associative, with the exception of assignment, which is right-associative, and the three comparison operations, which do not associate.

```

program ::= [[class;]]+
  class ::= class TYPE [inherits TYPE] { [[feature;]]* }
feature ::= ID( [ formal [, formal]]* ) : TYPE { expr }
  | ID : TYPE [ <- expr ]
formal ::= ID : TYPE
expr ::= ID <- expr
  | expr[@TYPE].ID( [ expr [, expr]]* )
  | ID( [ expr [, expr]]* )
  | if expr then expr else expr fi
  | while expr loop expr pool
  | { [[expr;]]+ }
  | let ID : TYPE [ <- expr ] [, ID : TYPE [ <- expr ]]* in expr
  | case expr of [[ID : TYPE => expr;]]+ esac
  | new TYPE
  | isvoid expr
  | expr + expr
  | expr - expr
  | expr * expr
  | expr / expr
  | ~ expr
  | expr < expr
  | expr <= expr
  | expr = expr
  | not expr
  | (expr)
  | ID
  | integer
  | string
  | true
  | false

```

Figure 1: Cool syntax.